
[bookmark: _Toc144509227]OData: составные элементы, операции и опции.
[bookmark: _Toc144509228]Используемые бизнес-объекты
Для целей демонстрации OData-элементов обозначим следующую модель данных – это контейнер переменных данных (TVARVC-like), а также логи, связанные с изменением данных.
С точки зрения базы данных имеется 3 таблицы:
1) Заголовочная таблица с переменными
2) Подчиненная таблица со значениями переменных
3) Таблица с логами по изменению заголовочных данных и по изменению значений переменных
А также связанное хранилище файлов с переменной. Так называемый справочный файл с описанием и пояснением.
Этих трех таблиц будет достаточно, чтобы показать возможности; а также, чтобы не усложнять понимание. А бинарные данные данные также будет использовать для показа возможностей OData в части чтения и загрузки файлов.
Продемонстрируем модель данных задачи без OData.
Транзакция ZWEB_ABAP - Web ABAP: manage records.
[image: ]
В транзакции ведения мы можем назначать ID переменной, делать описание, указывать как отдельное значение, так и RANGE-значений; использовать переменную в качестве SWITCH, а также подгружать к ней файлы и, что очень важно смотреть логи изменения по каждому полю.
Также есть возможность включать точку останова на переменную и делать ее неактивной для использования, не удаляя содержимого.

Для использования подобного функционала внутри бизнес-приложений достаточно написать подобное
	MO_VAR является инстанцией ZCL_WA001_VAR_READ.

	Пример использования переменных в любом бизнес приложении (для проверки на SWITCH)

	


	

	Чтение RANGE

	


	

	Чтение Separate Value

	



Пример работы внутри прикладного бизнес-приложения (клиентская программа)

Результат: в программе нет хардкодных значений; места использования можно найти через инструмент ведения и включения режима отладки.

Теперь посмотрим, как будет выглядеть модель этого же объекта, но уже в OData-проекте.

[bookmark: _Toc144509229]Представление бизнес-объекта в OData-проекте
OData-проект ведется (создается, просматривается и изменяется) – в транзакции SEGW.

 Тестировать сервис можно/нужно в транзакции /IWFND/GW_CLIENT - SAP Gateway Client. 
Составными частями информации про OData сервис являются:
1) Service document – набор сущностей, которые входят в конкретный web-Service
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/

	

	
Перечислены сущности со свойствами (аннотациями).
Имеется в каждом OData-сервисе.

2) Service metadata document – метаданные сервиса; это перечисление всех сущностей, с набором полей, Function Imports и другая подробная информация о свойствах сервисе (структура данных).
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/$metadata

	

	
Перечислены сущности со свойствами (аннотациями).
Имеется в каждом OData-сервисе.

Составные элементы проекта OData (SAP Gateway Service Builder):
1) Entity (Сущность) и Entity Type (тип сущности). Набор полей для отображения информации. (по сути представляет из себя структуру со свойствами). Свойства полей сущности контролируются аннотациями. Аннотации задаются в классе-модели сервиса (в этом примере: ZCL_ZWEB_ABAP_DEMO1_MPC + ZCL_ZWEB_ABAP_DEMO1_MPC_EXT ). MPC – model provider class.
Детально аннотации описаны здесь (какая аннотация за что отвечает): https://wiki.scn.sap.com/wiki/display/EmTech/SAP+Annotations+for+OData+Version+2.0

В каждом типе сущности есть ключ и он используется для ассоциаций (для связей между сущностями).

2) EntitySet (набор сущностей). Это массив сущностей, отношение 0:много. 
У EntitySet также есть свойства, и они также регулируются через аннотации.

3) Property (свойство/поле) – аналог поля таблицы или отдельно стоящей переменной.

4) Navigation Property (Свойство навигации).
Позволяет сделать переход от одной сущности сервиса к другой.
5) Association (Ассоциации). Определяет связь между сущности: кардинальность и поля, по которым идем связь.

В метаданных показано так

[bookmark: _Toc144509230]CRUD-Q операции в OData
Для работы с объектами доступны 5 базовых операций, перечисленные в таблице.
	Операций/действие над объектам
	OData-метод
	Маска ABAP-метод
на примере сущности VARHEADSET

	C – create / создать
	POST
	VARHEADSET_CREATE_ENTITY

	R – read / чтение записи
	GET (по ключу)
	VARHEADSET_GET_ENTITY

	U – update/ обновление
	PUT
PATCH
MERGE – устарел и более не рекомендован
	VARHEADSET_UPDATE_ENTITY

	D – delete/ удаление
	DELETE
	VARHEADSET_DELETE_ENTITY

	Q – query/ запрос списка
	GET (по $filter)
	VARHEADSET_GET_ENTITYSET



Не обязательно все опции должны быть реализованы (имплементированы), но те операции, которые реализованы

Рассмотрим операции и их минимальную реализацию на примере проекта ZWEB_ABAP_DEMO1.
Видим, что все операции проходят через класс ZCL_ZWEB_ABAP_DEMO1_DPC_EXT; именно в этом классе и реализуем в самых минимальных возможностях web-Service по протоколу OData.
[bookmark: _Toc144509231]Q – query / запрос списка
Запрос возвращает набор сущностей вызываемого типа

	GET

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet

	METHOD varheadset_get_entityset.
**TRY.
*CALL METHOD SUPER->VARHEADSET_GET_ENTITYSET
*  EXPORTING
*    IV_ENTITY_NAME           =
*    IV_ENTITY_SET_NAME       =
*    IV_SOURCE_NAME           =
*    IT_FILTER_SELECT_OPTIONS =
*    IS_PAGING                =
*    IT_KEY_TAB               =
*    IT_NAVIGATION_PATH       =
*    IT_ORDER                 =
*    IV_FILTER_STRING         =
*    IV_SEARCH_STRING         =
**    io_tech_request_context  =
**  IMPORTING
**    et_entityset             =
**    es_response_context      =
*    .
** CATCH /iwbep/cx_mgw_busi_exception .
** CATCH /iwbep/cx_mgw_tech_exception .
**ENDTRY.


    DATA lt_var_set TYPE zttwa001_var_h_srv.
    FIELD-SYMBOLS <fs_var_set> TYPE zswa001_var_h_srv.

    DATA lt_var_id TYPE ztwa001_varid_tab.
    DATA lt_var_val TYPE ztwa001_varval_tab.
    DATA lt_var_file TYPE ztwa001_varfile_tab.

    FIELD-SYMBOLS <fs_var_id> TYPE ztwa001_varid.
    FIELD-SYMBOLS <fs_var_list> TYPE zswa001_variable_alv.

    SELECT * FROM ztwa001_varid
        INTO TABLE lt_var_id
    "    WHERE var_name IN mo_dto_scr->mt_var_name_rng
        UP TO 1000 ROWS
        .

    IF lt_var_id IS NOT INITIAL.
      SELECT * FROM ztwa001_varval
         INTO TABLE lt_var_val
         FOR ALL ENTRIES IN lt_var_id
         WHERE  var_name EQ lt_var_id-var_name
         .

      SELECT * FROM ztwa001_varfile
        INTO TABLE lt_var_file
        FOR ALL ENTRIES IN lt_var_id
         WHERE  var_name EQ lt_var_id-var_name
         .
    ENDIF.


    CLEAR lt_var_set.

    LOOP AT lt_var_id ASSIGNING <fs_var_id>.
      APPEND INITIAL LINE TO lt_var_set ASSIGNING <fs_var_set>.
      MOVE-CORRESPONDING <fs_var_id> TO <fs_var_set>.

      <fs_var_set>-name = <fs_var_id>-var_name.
      <fs_var_set>-description = <fs_var_id>-var_desc.
      <fs_var_set>-var_type = <fs_var_id>-var_type.

      <fs_var_set>-is_del = <fs_var_id>-is_del.
      <fs_var_set>-debug_is_on = <fs_var_id>-is_debug_on.
      <fs_var_set>-fast_val = <fs_var_id>-fast_val.
      <fs_var_set>-cru = <fs_var_id>-cru.
      <fs_var_set>-crd = <fs_var_id>-crd.
      <fs_var_set>-crt = <fs_var_id>-crt.
      <fs_var_set>-chu = <fs_var_id>-chu.
      <fs_var_set>-chd = <fs_var_id>-chd.
      <fs_var_set>-cht = <fs_var_id>-cht.

      <fs_var_set>-num_of_values = REDUCE i( INIT vals_in_var = 0
                                              FOR ls_var_val IN lt_var_val WHERE ( var_name = <fs_var_id>-var_name )
                                              NEXT vals_in_var = vals_in_var + 1 ).

      <fs_var_set>-num_of_files = REDUCE i( INIT files_in_var = 0
                                              FOR ls_var_file IN lt_var_file WHERE ( var_name = <fs_var_id>-var_name )
                                              NEXT files_in_var = files_in_var + 1 ).
    ENDLOOP.

    """""""""""""""""""""""""""""""""""""""""""""""""""""""""""""""
    et_entityset[] = lt_var_set[].
    es_response_context-count = lines( et_entityset[] ).
    es_response_context-is_sap_data_exists_calculated = abap_true.
    "es_response_context-skiptoken


  ENDMETHOD.



Система возвращает не только данные, но и ссылки, по которым можно (и нужно) обращаться к сущности по ключу (то есть перейти от Query к Read).

Используя эту ссылку, мы можем прочитать конкретную запись.

[bookmark: _Toc144509232]R – read / чтение по ключу
Запросы возвращает одну сущность-структуру. Вызов происходит по ключу.

	GET

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZAPO_N15')
В скобках указывается ключ.

	METHOD varheadset_get_entityset.
*        importing
*      !IV_ENTITY_NAME type STRING
*      !IV_ENTITY_SET_NAME type STRING
*      !IV_SOURCE_NAME type STRING
*      !IT_KEY_TAB type /IWBEP/T_MGW_NAME_VALUE_PAIR
*      !IO_REQUEST_OBJECT type ref to /IWBEP/IF_MGW_REQ_ENTITY optional
*      !IO_TECH_REQUEST_CONTEXT type ref to /IWBEP/IF_MGW_REQ_ENTITY optional
*      !IT_NAVIGATION_PATH type /IWBEP/T_MGW_NAVIGATION_PATH
*    exporting
*      !ER_ENTITY type ZCL_ZWEB_ABAP_DEMO1_MPC=>TS_VARHEAD
*      !ES_RESPONSE_CONTEXT type /IWBEP/IF_MGW_APPL_SRV_RUNTIME=>TY_S_MGW_RESPONSE_ENTITY_CNTXT
*    raising
*      /IWBEP/CX_MGW_BUSI_EXCEPTION
*      /IWBEP/CX_MGW_TECH_EXCEPTION .

    DATA lr_name_value_line TYPE REF TO /iwbep/s_mgw_name_value_pair.
    DATA ls_srv_in TYPE zswa001_var_h_srv.
    DATA ls_srv_out TYPE zswa001_var_h_srv.

    DATA lt_var_id TYPE ztwa001_varid_tab.
    DATA lt_var_val TYPE ztwa001_varval_tab.
    DATA lt_var_file TYPE ztwa001_varfile_tab.

    FIELD-SYMBOLS <fs_var_id> TYPE ztwa001_varid.
    FIELD-SYMBOLS <fs_var_list> TYPE zswa001_variable_alv.

    LOOP AT it_key_tab REFERENCE INTO lr_name_value_line.
      CASE lr_name_value_line->name.
        WHEN 'NAME' OR 'Name'.
          ls_srv_in-name = lr_name_value_line->value.
        WHEN OTHERS.

      ENDCASE.
    ENDLOOP.

    SELECT * FROM ztwa001_varid
        INTO TABLE lt_var_id
        WHERE var_name = ls_srv_in-name.

    SELECT * FROM ztwa001_varval
        INTO TABLE lt_var_val
        WHERE var_name = ls_srv_in-name.

    LOOP AT lt_var_id ASSIGNING <fs_var_id>.

      MOVE-CORRESPONDING <fs_var_id> TO ls_srv_out.

      ls_srv_out-name = <fs_var_id>-var_name.
      ls_srv_out-description = <fs_var_id>-var_desc.
      ls_srv_out-var_type = <fs_var_id>-var_type.

      ls_srv_out-is_del = <fs_var_id>-is_del.
      ls_srv_out-debug_is_on = <fs_var_id>-is_debug_on.
      ls_srv_out-fast_val = <fs_var_id>-fast_val.
      ls_srv_out-cru = <fs_var_id>-cru.
      ls_srv_out-crd = <fs_var_id>-crd.
      ls_srv_out-crt = <fs_var_id>-crt.
      ls_srv_out-chu = <fs_var_id>-chu.
      ls_srv_out-chd = <fs_var_id>-chd.
      ls_srv_out-cht = <fs_var_id>-cht.

      ls_srv_out-num_of_values = REDUCE i( INIT vals_in_var = 0
                                              FOR ls_var_val IN lt_var_val WHERE ( var_name = <fs_var_id>-var_name )
                                              NEXT vals_in_var = vals_in_var + 1 ).

      ls_srv_out-num_of_files = REDUCE i( INIT files_in_var = 0
                                              FOR ls_var_file IN lt_var_file WHERE ( var_name = <fs_var_id>-var_name )
                                              NEXT files_in_var = files_in_var + 1 ).

      EXIT.
    ENDLOOP.

    MOVE-CORRESPONDING ls_srv_out TO er_entity.

    IF ls_srv_out IS INITIAL.
      es_response_context-no_content = abap_true.
    ENDIF.




Структуру, которую система вернула по запросу GET (by key) можно использовать как образец для CREATE / UPDATE- запросов с помощью кнопки Use as Request.

[bookmark: _Toc144509233]C – create / создание записи
При создании используем метод POST и указываем имя Set.

В случае успеха система пришлет статус 201. С точки зрения ABAP-реализации этот статус означает, что не возникло исключительных ситуаций. Не факт, что данные обновились и/или корректно обновились.

	POST

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet

	Тело запроса

	<?xml version="1.0" encoding="utf-8"?>
<entry xml:base="http://vhcalnplci:8000/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/" xmlns="http://www.w3.org/2005/Atom" xmlns:m="http://schemas.microsoft.com/ado/2007/08/dataservices/metadata" xmlns:d="http://schemas.microsoft.com/ado/2007/08/dataservices">
 <id>http://vhcalnplci:8000/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZAPO_N15')</id>
 <title type="text">VarHeadSet('ZAPO_N15')</title>
 <updated>2021-05-13T20:39:35Z</updated>
 <category term="ZWEB_ABAP_DEMO1_SRV.VarHead" scheme="http://schemas.microsoft.com/ado/2007/08/dataservices/scheme"/>
 <link href="VarHeadSet('ZAPO_N15')" rel="self" title="VarHead"/>
 <link href="VarHeadSet('ZAPO_N15')/VarID2Values" rel="http://schemas.microsoft.com/ado/2007/08/dataservices/related/VarID2Values" type="application/atom+xml;type=feed" title="VarID2Values"/>
 <content type="application/xml">
  <m:properties>
   <d:Name>ZAPO_N15</d:Name>
   <d:Description>APO key Customers Number</d:Description>
   <d:VarType>3</d:VarType>
   <d:VarTypeTx/>
   <d:NumOfValues>3</d:NumOfValues>
   <d:NumOfFiles>0</d:NumOfFiles>
   <d:IsDel>false</d:IsDel>
   <d:DebugIsOn>false</d:DebugIsOn>
   <d:FastVal/>
   <d:Cru>DEVELOPER</d:Cru>
   <d:Crd>2021-05-11T00:00:00</d:Crd>
   <d:Crt>PT22H12M38S</d:Crt>
   <d:Chu>DEVELOPER</d:Chu>
   <d:Chd>2021-05-12T00:00:00</d:Chd>
   <d:Cht>PT14H08M02S</d:Cht>
  </m:properties>
 </content>
</entry>

	  METHOD varheadset_create_entity.
*        importing
*      !IV_ENTITY_NAME type STRING
*      !IV_ENTITY_SET_NAME type STRING
*      !IV_SOURCE_NAME type STRING
*      !IT_KEY_TAB type /IWBEP/T_MGW_NAME_VALUE_PAIR
*      !IO_TECH_REQUEST_CONTEXT type ref to /IWBEP/IF_MGW_REQ_ENTITY_C optional
*      !IT_NAVIGATION_PATH type /IWBEP/T_MGW_NAVIGATION_PATH
*      !IO_DATA_PROVIDER type ref to /IWBEP/IF_MGW_ENTRY_PROVIDER optional
*    exporting
*      !ER_ENTITY type ZCL_ZWEB_ABAP_DEMO1_MPC=>TS_VARHEAD
*    raising
*      /IWBEP/CX_MGW_BUSI_EXCEPTION
*      /IWBEP/CX_MGW_TECH_EXCEPTION .
    DATA ls_srv_in TYPE zcl_zweb_abap_demo1_mpc=>ts_varhead.

    io_data_provider->read_entry_data( IMPORTING es_data = ls_srv_in ).
    MOVE-CORRESPONDING ls_srv_in TO er_entity.

  ENDMETHOD.



Видим, что важным (и довольно понятным с точки зрения использования в своей реализации) является чтение входных параметров через
 io_data_provider->read_entry_data( IMPORTING es_data = ls_srv_in ).

[bookmark: _Toc144509234]U – update/ обновление записи
При создании указываем метод PUT или PATCH, а также ключ записи. Успешным ответом является 204. Ответ ( response) приходит без данных.

	PUT – при указании тех полей, которые требуется обновить (не обязательно указывать все поля).
PATCH – в случае указания части поле система сначала прочитает запись через GET_ENTITY (READ), то есть дообогатит запись; а потом запустит метод UDPATE. 

Можно переопределить поведение через method redifinition
/IWBEP/IF_MGW_APPL_SRV_RUNTIME~PATCH_ENTITY

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZAPO_N15')

	Тело запроса

	<?xml version="1.0" encoding="utf-8"?>
<entry xml:base="http://vhcalnplci:8000/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/" xmlns="http://www.w3.org/2005/Atom" xmlns:m="http://schemas.microsoft.com/ado/2007/08/dataservices/metadata" xmlns:d="http://schemas.microsoft.com/ado/2007/08/dataservices">
 <id>http://vhcalnplci:8000/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZAPO_N15')</id>
 <title type="text">VarHeadSet('ZAPO_N15')</title>
 <updated>2021-05-13T20:39:35Z</updated>
 <category term="ZWEB_ABAP_DEMO1_SRV.VarHead" scheme="http://schemas.microsoft.com/ado/2007/08/dataservices/scheme"/>
 <link href="VarHeadSet('ZAPO_N15')" rel="self" title="VarHead"/>
 <link href="VarHeadSet('ZAPO_N15')/VarID2Values" rel="http://schemas.microsoft.com/ado/2007/08/dataservices/related/VarID2Values" type="application/atom+xml;type=feed" title="VarID2Values"/>
 <content type="application/xml">
  <m:properties>
   <d:Name>ZAPO_N15</d:Name>
   <d:Description>APO new text</d:Description>
  </m:properties>
 </content>
</entry>

	  METHOD varheadset_update_entity.
**TRY.
*CALL METHOD SUPER->VARHEADSET_UPDATE_ENTITY
*  EXPORTING
*    IV_ENTITY_NAME          =
*    IV_ENTITY_SET_NAME      =
*    IV_SOURCE_NAME          =
*    IT_KEY_TAB              =
**    io_tech_request_context =
*    IT_NAVIGATION_PATH      =
**    io_data_provider        =
**  IMPORTING
**    er_entity               =
*    .
** CATCH /iwbep/cx_mgw_busi_exception .
** CATCH /iwbep/cx_mgw_tech_exception .
**ENDTRY.

    DATA ls_srv_in TYPE zcl_zweb_abap_demo1_mpc=>ts_varhead.

    io_data_provider->read_entry_data( IMPORTING es_data = ls_srv_in ).
    MOVE-CORRESPONDING ls_srv_in TO er_entity.

  ENDMETHOD.



[bookmark: _Toc144509235]D – delete / удаление записи
При создании указываем метод DELETE, а также ключ записи. Успешным ответом является 204. Ответ ( response) приходит без данных.
[image: ]

	DELETE 

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZAPO_N15')

	  METHOD varheadset_delete_entity.
*    importing
*      !IV_ENTITY_NAME type STRING
*      !IV_ENTITY_SET_NAME type STRING
*      !IV_SOURCE_NAME type STRING
*      !IT_KEY_TAB type /IWBEP/T_MGW_NAME_VALUE_PAIR
*      !IO_TECH_REQUEST_CONTEXT type ref to /IWBEP/IF_MGW_REQ_ENTITY_D optional
*      !IT_NAVIGATION_PATH type /IWBEP/T_MGW_NAVIGATION_PATH
*    raising
*      /IWBEP/CX_MGW_BUSI_EXCEPTION
*      /IWBEP/CX_MGW_TECH_EXCEPTION .

    DATA lr_name_value_line TYPE REF TO /iwbep/s_mgw_name_value_pair.
    DATA ls_srv_in TYPE zcl_zweb_abap_demo1_mpc=>ts_varhead.

    LOOP AT it_key_tab REFERENCE INTO lr_name_value_line.
      CASE lr_name_value_line->name.
        WHEN 'NAME' OR 'Name'.
          ls_srv_in-name = lr_name_value_line->value.
        WHEN OTHERS.

      ENDCASE.
    ENDLOOP.

  ENDMETHOD.



Мы рассмотрели базовые операции и их минимальную реализацию; а теперь рассмотрим базовые опции.
[bookmark: _Toc144509236]Использование Navigation Property и inlining при помощи $expand
Navigation Property – это свойство, обеспечивающее связь между одной и другой сущностью. 
В SEGW-проекте (GateWay-проект) его можно наблюдать в описаниях свойства сущностей
[image: ]
В metadata

Использование Navigation Property позволяет вернуть (выбрать) данные по параметрам вышестоящей сущности через NavigationProperty.
В нашем случае, NavigationProperty VarID2Values, который находится в сущности VarHeadSet. Пример запроса с результатом ниже
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZAPO_N15')/VarID2Values

	Система вернет только записи нижестоящей сущности

	


	



Однако с помощью опции $expand и NavigationProperty можно вернуть вложенную структуру (Deep Entity).
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZAPO_N15')?$expand=VarID2Values

	Система вернет записи ведущей сущности (principal entity) и зависимой сущности (dependent entity).

	


	



Совместное отображение информации – называется inline-отображением. И в целом, термин inline – обозначает отображение /обработку не в отдельном блоке, а в блоке «по требованию»; использование чего «на лету», «встроено».

[bookmark: _Toc144509237]Опции при выборке
OData позволяет клиенту указывать серверу параметры по которым нужно сделать ограничение (критерии отбора), набор полей для выборки, параметры пагинации (ограничение выборки сверху и снизу), сортировки полей и некоторые другие. Такие параметра называются опциями (OData Query Options). Они применимы, прежде всего, для выборок (то есть методов GET Entity Set). Одну из опций мы уже видели – это $expand. Собственно, опции используются со специальным символом $ в запросе.
Применимость опций представлено в таблице с примерами ниже. Обращаю внимание, что это способ передачи параметров, а не их обработки внутри выборок. Как именно будут обрабатываться (и будут ли вообще) – зависит от конкретной задачи.
	OData-option
	Комментарий

	$filter
	Для передачи параметров по ограничению по полям (которые находятся в OData-сущности).

	$search
	Передача строки для поиска без привязки к свойствам Entity.
Обращаю внимание, что в строке передаётся без символа $.

	$select
	Передача полей с клиента на сервер, которые необходимо вернуть в ответ.

	$orderby
	Передача полей и направление сортировки.

	$inlinecount
	Подсчет количества записей по выборке и встраивание этого количества в результирующую выборку. Используется для пагинации (paging) совместно с $top и $skip.

	$count
	Возвращает количество записей, но в отличие от $inlinecount в результирующей выборке возвращается только количество записей без других данных.

	$expand
	Позволяет получить расширенную сущность: Ведущая сущность + подчиненная сущность.

	$format
	Возвращаемый формат данных

	$top и $skip
	Пагинация средствами клиента (то есть UI5-приложение указывает, какую порцию (page) данных нужно возвратить).

	$skiptoken
	Сервер формирует порцию (page) и указывает номер следующей записи.



[bookmark: _Toc144509238]$filter – ограничение выборки
Указываем в дополнение к запросу GetEntitySet. Предназначается для передачи параметров фильтрации. Можно или нельзя фильтровать по определенному параметру указывается в аннотациях ([sap: filterable]).
Данную аннотацию можно указать через свойства. Столбец Filterable.
[image: ]

На уровне Set можно установить, что он Require Filters
[image: ]

В $metadata в случае, если свойство Filterable – оно отображаться не будет. Будет отображаться только если свойство = false.
[image: ]

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$filter=(Name eq 'ZSD_N1')

	    " $filter - reading
    DATA lo_req_filter TYPE REF TO /iwbep/if_mgw_req_filter.
    DATA lt_select_options TYPE /iwbep/t_mgw_select_option.
    DATA lt_varid_range TYPE RANGE OF ztwa001_varid-var_name.

    FIELD-SYMBOLS <fs_select_option> TYPE /iwbep/s_mgw_select_option.

    lo_req_filter ?= io_tech_request_context->get_filter( ).
    lt_select_options = lo_req_filter->get_filter_select_options(  ).

    LOOP AT lt_select_options ASSIGNING <fs_select_option>.
      CASE <fs_select_option>-property.
        WHEN 'Name' or 'NAME'.
          lo_req_filter->convert_select_option( EXPORTING is_select_option = <fs_select_option>
                                                IMPORTING et_select_option = lt_varid_range ).


        WHEN OTHERS.

      ENDCASE.
    ENDLOOP.

	


	

	

	В $filter можно перечислять несколько свойств. Однако OData в данном случае будет отвечать лишь за передачу значений из сервиса в ABAP, а вот как это уже будет использовано внутри – полностью на усмотрение конкретной задачи и разработчика/консультанта.

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$filter=(Name eq 'ZAPO_N42' and Description eq 'ZSD_N1')

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$filter=(Name eq 'ZAPO_N42' or Name eq 'ZSD_N1')

	

	

	

	В $filter можно указывать также и маску с помощью substringof / startswith / endswith

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$filter=(substringof('APO', Name))

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$filter=(startswith(Name, 'ZSD_N1'))

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$filter=(endswith(Name, 'ZSD_N1'))

	

	

	А вот при комбинации параметров (или при усложненных параметров выборки) возможно придется разбирать уже саму строку.

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$filter=(Name eq 'ZSD_N1' or (substringof('Sales', Description)))

	


	

	Доступны операторы: gt, lt, eq, ne, startswith, endswith, substring
Недоступны: in, 
А также custom options




[bookmark: _Toc144509239]$search – передача строки поиска
Указываем в дополнение к запросу GetEntitySet. Можно или нельзя передавать этот параметр указывается в аннотациях ([sap: searchable]).
Данную аннотацию можно указать через свойства. Столбец Searchable.
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?search=do the needful

	!!! Опция передается без символа $

	В ABAP читается из параметра IV_SEARCH_STRING

	


	




[bookmark: _Toc144509240]$select – выбор нужных полей с данными (чтобы получить список без излишеств).
Клиент может указать серверу, какие именно поля необходимо вернуть в запросе. Такой праметра можно использовать для улучшение производительности, а также для подсчета агрегаций (какие поля передали – по тем и посчитали агрегированные значения).
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$select=Name,Description,VarType

	    " $select - option: перечисление полей
    DATA lt_select_fields TYPE /iwbep/t_mgw_tech_field_names.
    lt_select_fields = io_tech_request_context->get_select( ).

	


	




[bookmark: _Toc144509241]$inlinecount – передача необходимости подсчета общего количества записей
Указываем в дополнение к запросу, чтобы сервер вернул количество записей, которое отвечает критериям выборки. Параметр $inlinecount может принимать 2 значения: allpages и None.
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$inlinecount=allpages

	    " $inlinecount=allpages
    IF io_tech_request_context->has_inlinecount( ) EQ abap_true.
      es_response_context-inlinecount = select_from_tab_with_count(  ).
    ENDIF.

	


	




[bookmark: _Toc144509242]$count – только количество строк в результирующей выборке 
Указываем в дополнение к запросу GetEntitySet. В отличие от $inlinecount в результирующей выборке будет только количество строк.
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet/$count

	    IF io_tech_request_context->has_count( ) EQ abap_true.
      es_response_context-count = lines( et_entityset[] ).
    ENDIF.

	


	




[bookmark: _Toc144509243]$orderby – передача параметров сортировки с клиента на сервер (Server-Side Sorting).
Параметры для сортировки: набор полей и направление сортировки. 
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$orderby=Description desc,VarType asc

	    " $orderby - поля для сортировки
    DATA lt_orderby_fields TYPE /iwbep/t_mgw_tech_order.
    lt_orderby_fields = io_tech_request_context->get_orderby( ).

	


	Как правило, используется совместно с $top / $skip для Client Side Paging




[bookmark: _Toc144509244]$expand – встраивание (inlining) дополнительных данных в выборку 
Указываем в дополнение к запросу GetEntitySet. Позволяет получить расширенную сущность Deep Entity типа Заголовок-Позиция.
Внутри ABAP-реализации есть две возможности:
1) По умолчанию: система сделает выборку ведущей сущности (в нашем случае VarHeadSet, а потом по каждой записи будет вызывать метод GetEntity для подчиненной сущности).
Иногда это может быть приемлемо, а иногда это может быть слишком затратно по производительности. 

2) Реализация через метод Get_Expanded_Entity / Get_Expanded_Entity_Set.
/IWBEP/IF_MGW_APPL_SRV_RUNTIME~GET_EXPANDED_ENTITY
/IWBEP/IF_MGW_APPL_SRV_RUNTIME~GET_EXPANDED_ENTITYSET

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZAPO_N15')?$expand=VarID2Values

	Система вернет записи ведущей сущности (principal entity) и зависимой сущности (dependent entity).

	В случае реализации по умолчанию (вариант 1) нужно иметь ввиду параметр IV_SOURCE_NAME для зависимой сущности.


	


	


	

	

	





[bookmark: _Toc144509245]$format – формат возвращаемых данных. xml json xlsx
Указываем в дополнение практически к любому запросу. В стандарте OData предполагает только два формата (xml+json), но SAP реализовал и еще один.
Параметр XML – используется по умолчанию. Считается, что этот формат является более удобным для чтения человеком. А вот формат JSON более предпочтителен
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZAPO_N15')?$format=json

	Этот параметр никак не обрабатывается в ABAP-реализации OData-operations, а обрабатывается автоматически NetWeaver GateWay.

	


	




[bookmark: _Toc144509246]$top $skip – пагинация средствами клиента (client-side pagination) 
Указываем в дополнение к запросу GetEntitySet. Можно или нельзя передавать этот параметр указывается в аннотациях ([sap: searchable]).
Данную аннотацию можно указать через свойства. Столбец Searchable.
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$top=10&$skip=20

	$top – дает указание серверу, сколько нужно вернуть  записей (точнее верхнее ограничение по количество)
$skip – параметр, указыающий, сколько записей нужно пропустить при выборке.
Как правило, их сопровождает параметр $orderby, чтобы было явное понимание о том в каком порядке располагать записи, чтобы их пропустить. 

	    " $top / $skip
    " is_paging- передается явно
    data(lv_top_records) = io_tech_request_context->get_top( ).
    data(lv_skip_offset) = io_tech_request_context->get_skip( ).

    SELECT * FROM ztwa001_varid
         WHERE var_name IN @lt_varid_range ORDER BY PRIMARY KEY
        INTO TABLE @lt_var_id
        OFFSET @is_paging-skip
        UP TO @is_paging-top ROWS.

	


	




[bookmark: _Toc144509247]$skiptoken – пагинация средствами сервера (server-side pagination) 
Параметр позволяет выдавать данные с сервера порционно; при этом пагинацию контролируют сервер, а не клиент.
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$skiptoken=8

	$skiptoken – указание серверу, с какой записи стоит начать отображение. Размер страницы (аналог параметра $top – сервер при этом определяет сам).
Сервер в параметре es_response_context-skiptoken должен вернуть либо пусто (если следующей порции нет, либо номер следующей записи).

	    " $skiptoken - пагинации средствами сервера
    DATA lv_token_size_on_server TYPE syindex VALUE 12.
    DATA lv_token_size_on_server_plus10 TYPE syindex.
    DATA lv_skiptoken_step TYPE sytabix.
    DATA one_due2minus_offset TYPE syindex VALUE 1.
    DATA one_due2next_record TYPE syindex VALUE 1.

    lv_skiptoken_step = io_tech_request_context->get_skiptoken( ).
    " skiptoken - это не просто skip, а начало следующей записи, поэтому - для OFFSET
    lv_skiptoken_step = lv_skiptoken_step - 1.

    " чтобы рассчитать
    lv_token_size_on_server_plus10 = lv_token_size_on_server + 10.
    SELECT * FROM ztwa001_varid
     WHERE var_name IN @lt_varid_range ORDER BY PRIMARY KEY
    INTO TABLE @lt_var_id
    OFFSET @lv_skiptoken_step
    UP TO @lv_token_size_on_server_plus10 ROWS.

    IF lines( lt_var_id ) GT lv_token_size_on_server.
      DELETE lt_var_id FROM lv_token_size_on_server.
      es_response_context-skiptoken = lv_skiptoken_step + lv_token_size_on_server + one_due2minus_offset + one_due2next_record.
      CONDENSE es_response_context-skiptoken NO-GAPS.
    ELSE.
      CLEAR es_response_context-skiptoken.
    ENDIF.

	


	




[bookmark: _Toc72796760][bookmark: _Toc144509248]Массовая отправка запросов (batch-request) – принцип работы
Batch-запросы используются, чтобы объединить несколько обращений с клиента к серверу в один запрос.
batch-запрос – это по сути объединение нескольких вызовов в один «большой» вызов. В Batch-запросах можно комбинировать запросы на чтение и запись, а также использовать различные сущности. При запросах на обновление (CUD) для соблюдения LUW – нужно использовать соответствующий change-sets (в этом случае COMMIT или ROLLBACK можно будет применить ко всем данным).
Давайте рассмотрим, как выглядит простой batch-request для операции чтения, чтобы понимать принцип и назначение batch-request.
Запросы batch всегда выполняются через метод POST. Но могут включать в себя разные операции CRUDQ.
[image: ]

При выполнении batch в теле запроса нужно указывать:
1) наименование batch-запроса (так называемые границы). Для GET/GetSet – change_set указывать не нужно; для CREATE/UPDATE/DELETE – указывать обязательно.
2) тело Batch-запроса должно состоять из явных мини-запросов с правильным указанием количества отступов.
	Пример BODY-batch-запроса
Между мини-запросами – должен быть отступ минимум 2 строки

	HTTP-method 
	POST

	Request URI
	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/$batch

	Content-Type
	multipart/mixed;boundary=batch_wababap01

	Ниже отправляются такие запросы:

/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZAPO_N15')?$expand=VarID2Values
/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZFI_N2')
/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet('ZAPO_N6')

	--batch_wababap01
Content-Type: application/http
Content-Transfer-Encoding: binary

GET VarHeadSet('ZAPO_N15')?$expand=VarID2Values HTTP/1.1


--batch_wababap01
Content-Type: application/http
Content-Transfer-Encoding: binary

GET VarHeadSet('ZFI_N2') HTTP/1.1



--batch_wababap01
Content-Type: application/http
Content-Transfer-Encoding: binary

GET VarHeadSet('ZAPO_N6') HTTP/1.1


--batch_wababap01--



3) в строке подключения должен быть $batch
	Пример строки (адреса) запроса

	/sap/opu/odata/SAP/ZWEB_ABAP_DEMO3_SRV/$batch



4) в заголовке запроса должен быть Content-Type с указанием boundary
	Заголовок запроса

	Content-Type: multipart/mixed;boundary=batch_wababap01

	




Это бы пример batch на чтение. Нам не требовалось указание change-set. 
При выполнении система единожды отправила запрос в NetWeaver Gateway, но выполняла их последовательно так, как если бы они были отправлены один за одним. 
В качестве результата вернулся объединенный результат всех запросов.
В блоке OData на основе композиции мы разберем пример и на изменение (либо можно его прямо сейчас просмотреть в том блоке, но лучше попорядку).

[bookmark: _Toc144509249]Метод DEFINE в *MPC для указания дополнительных аннотаций и свойств
Для метаданных и обработки данных из OData на стороне клиента играют аннотации. Иногда корректировки свойств через SEGW недостаточно, чтобы корректно настроить metadata сервиса. Для целей более гибкого управления метаданными можно/нужно использовать переопределение метода DEFINE класса ZCL_ZWEB_ABAP_DEMO1_MPC через ZCL_ZWEB_ABAP_DEMO1_MPC_EXT.
Случаи, для которых может потребоваться правка аннотаций через переопределение: корректировка формата отображения, указания свойства Media, указать зависимости между свойствами и множество других. Аннотации позволяют клиенту «понимать» как следует взаимодействовать с тем или иным полем, а также строить автоматически дополнительные элементы к полю (например, средство поиска).
	Класс ZCL_ZWEB_ABAP_DEMO1_MPC_EXT, метод DEFINE

	https://blogs.sap.com/2016/03/17/code-snippet-for-adding-annotations-to-your-gateway-service/

	    DATA lo_entity_type TYPE REF TO /iwbep/if_mgw_odata_entity_typ.
    DATA lo_property    TYPE REF TO /iwbep/cl_mgw_odata_property.
    DATA lo_annotation  TYPE REF TO /iwbep/if_mgw_odata_annotation.

    lo_entity_type = model->get_entity_type( iv_entity_name = 'VarHead').
    lo_property ?= lo_entity_type->get_property( iv_property_name = 'VarType').
    lo_annotation ?= lo_property->/iwbep/if_mgw_odata_annotatabl~create_annotation(
        iv_annotation_namespace = /iwbep/if_mgw_med_odata_types=>gc_sap_namespace ).

    lo_annotation->add( iv_key = 'display-format' iv_value = 'value-list' ).
    """"""""""""""""""""""""""""""""""""""""""""""""""""""""""""""
    lo_property ?= lo_entity_type->get_property( iv_property_name = 'VarTypeTx').
    lo_annotation ?= lo_property->/iwbep/if_mgw_odata_annotatabl~create_annotation(
        iv_annotation_namespace = /iwbep/if_mgw_med_odata_types=>gc_sap_namespace ).
    lo_annotation->add( iv_key = 'text-for' iv_value = 'VarType' ).

Корректно указанные аннотации могут существенно ускорить разработку на стороне клиента, а также дают возможность управлять свойствами через Gateway.

[bookmark: _Toc144509250]Debug-Mode и Трассировка вызовов OData-запросов
[bookmark: _Toc144509251]sap-ds-debug=true : расширенный технический режим

Проверять OData-запросы можно не только через Gateway-клиент (как мы делали выше), но и, прежде всего, через браузер.
Например, запрос на $skiptoken мог бы выглядеть так
	https://abap2020.space:44300/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$skiptoken=8

	




При проверке запросов мы можем добавить опцию ?sap-ds-debug=true и тогда система отобразит больше технической информации, в том числе ABAP-объект, который возвращает данные.
	https://abap2020.space:44300/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/VarHeadSet?$skiptoken=8&sap-ds-debug=true

	


	



Как видно, показано еще и время, а также подробные данные request и response.
Из Client Gateway также можно отображать этот режим; данные будут в таков виде:
[image: ]

[bookmark: _Toc144509252]Трассировка запрос и логирование ошибок
Для получения еще более подробной информации можно использовать трассировку OData-вызовов с помощью транзакции /IWFND/TRACES - SAP Gateway Traces. 
Трассировка активируется для пользователя
[image: ]

Затем под пользователем нужно выполнить действия. Система зафиксирует все запросы, которые были под пользователем.
На вкладке Performance Trace будет отображена информация о том, какие ABAP-объекты участвовали в обработке ответа.
[image: ]
На вкладке Payload Trace будет отображена информация о параметрах вызова и тот ответ, который был возвращен.
[image: ]

Будет отображен запрос с параметрами (в том числе body, если это POST, например)
[image: ]
И ответ, который был возвращен со стороны OData.
[image: ]

Более того, запрос на указанных параметрах, можно «перевести» в режим Gateway_client и уже детально протрассировать в изолированном режиме.
[image: ]

В случае, если возникает критичная (красная или приближенная к дампу) ошибка при вызове какого-либо OData-сервиса, то, как правило, ее можно найти в транзакциях: /IWFND/ERROR_LOG - SAP Gateway Error Log и/или /IWBEP/ERROR_LOG - SAP GW Backend Error Log. Использование этих транзакции может существенно ускорить понимание того, что «пошло не так». 
Эти транзакции логируют информацию без дополнительного включения трассировки.
Вид транзакции /IWFND/ERROR_LOG 
[image: ]

[bookmark: _Toc144509253]Тестирование через внешние инструменты
[bookmark: _Toc144509254]Postman
Упомянем немного и внешние инструменты (тоже очень полезные). 
https://learning.postman.com/docs/writing-scripts/test-scripts/
В postman создан для тестирования и это уже не просто инструмент тестирования, а API-рабочая среда. Есть бесплатная версия, есть платная версия.
Давайте рассмотрим пример по отправке запроса.

А также по автоматизации запросов + нагрузочному тесту.
Обратить внимание на ведение переменных в коллекции тестов.

[image: ]

[bookmark: _Toc144509255]Python -библиотека aiohttp
https://docs.aiohttp.org/en/stable/index.html
В рамках данной части курса не будем рассматривать, так как это отдельная тема: построение самописного приложения для тестирования. Иногда это имеет смысл; иногда – может и не иметь смысла. По ссылке сама библиотека хорошо задокументирована с множеством примеров.

[bookmark: _Toc144509256]Заключение по разделу
Мы рассмотрели общее представление о реализации web-service по протоколу OData в SAP NetWeaver Gateway. Также узнали, какие опции можно использовать в OData и как их использовать в ABAP-объектах. В разделе также были затронуты возможности трассировки и просмотра ошибок с их основными возможностями. По ходу раздела были приведены ссылки на различные источники информации и транзакции, которые собраны в таблице ниже.

	Полезные ссылки

	
	

	GWFND – sap help
	https://help.sap.com/viewer/68bf513362174d54b58cddec28794093/7.51.9/en-US/7db1ea508f88bb7ee10000000a445394.html

	OData – developer guide
	https://help.sap.com/viewer/68bf513362174d54b58cddec28794093/202009.001/en-US/a6422751c639276ee10000000a445394.html

	$skiptoken option
	https://blogs.sap.com/2020/10/23/skiptoken-system-query-option-in-sap-odata/

	MS Doc
OData protocol
	https://docs.microsoft.com/en-us/openspecs/windows_protocols/ms-odata/2b686a1a-9e1f-456f-80ff-072a010fc278

	Аннотации
	https://wiki.scn.sap.com/wiki/display/EmTech/SAP+Annotations+for+OData+Version+2.0

	Msdn-query optionss
	https://docs.microsoft.com/en-us/odata/concepts/queryoptions-overview

	Display-format
	https://blogs.sap.com/2016/03/17/code-snippet-for-adding-annotations-to-your-gateway-service/

	OData-troubleshooting
	https://blogs.sap.com/2018/09/09/gateway-odata-service-troubleshooting-and-detailed-tracing-in-abap/

	
	



	Важные транзакции

	SEGW - SAP Gateway Service Builder

	/IWFND/GW_CLIENT - SAP Gateway Client

	/IWFND/TRACES - SAP Gateway Traces

	/IWFND/ERROR_LOG - SAP Gateway Error Log

v<feed xmlns="http://www.w3.0org/2005/Atom" xmlns:m="http://schemas.microsoft.com/ado/2007/08/dataservices/metadata” xmlns:d="http:
xml:base="https://abap2020.space:44300/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1_SRV/">
<id>https://abap2020.space:44300/sap/opu/odata/SAP/ZWEB_ABAP_DEMO1l_SRV/VarHeadSet</id>

<link href="VarHeadSet('')" rel="self" title="VarHead"/>
<link href="VarHeadSet('')/VarID2Values" rel="http://schemas.microsoft.com/ado/2007/08/dataservices/related/VarID2Values" type
v<content type="application/xml">
v<m:properties xmlns:m="http://schemas.microsoft.com/ado/2007/08/dataservices/metadata” xmlns:d="http://schemas.microsoft.con

